Problem 12

What is the value of the first triangle number to have over five hundred divisors?

Solution

The triangle numbers are $\frac{1}{2} n(n+1)$ as n varies. The number of divisors of a number $p_1^{a_1} p_2^{a_2} \quad p_n^{a_n}$ is simply $(a_1+1)(a_2+1) \quad (a_n+1)$, since a divisor can only have the p_i as its prime factors, and it can have any of 0 to a_1 of p_1, etc. So we create a function which outputs the number of divisors of the nth triangular number, noting that n and $n+1$ share no factors, and that whichever is even will have one removed from the power of 2. I have allowed myself access to the function FactorInteger, because its arguments are small enough that a naïve implementation would be sufficient.

```
In[57]:= f[1] = 1;
f[2] = 2;
f[n_] := With[{even = First@Select[{n, n + 1}, EvenQ, 1],
    odd = First@Select[{n, n + 1}, OddQ, 1]}, Times @@ (1 + Last /@ FactorInteger[odd])
    Times @@ (1 + Last /@ FactorInteger[even / 2])]
SetAttributes[f, Listable]
```

Test:

```
In[44]:= Block[{n = 10}, Length[Divisors[n (n + 1)/2]] - f[n]]

Out[44]= 0
```

The function takes not very much time:

```
In[62]:= f[Range[10 000]]; // AbsoluteTiming

Out[62]= {0.531188, Null}
```

But more than built-in *Mathematica* does:

```
In[56]:= Function[{n}, Length[Divisors[n (n + 1)/2]]] /@ Range[10 000]; // AbsoluteTiming

Out[56]= {0.253493, Null}
```

Anyway, it gets the answer:

```
In[66]:= Module[{n = 2}, While[f[n] < 500, n++]; n (n + 1) / 2] // AbsoluteTiming

Out[66]= {0.545774, 76 576 500}
```

We could, however, have safely assumed that n would not be prime, and nor would $n+1$. We have also an upper bound: we know that $2^9 == 512$, so the product of the first nine primes will have enough divisors; we can turn this into a likely upper bound by finding the triangle number below it, since 2^9 has some slack in the number of divisors, and since it's getting quite inefficient already (the ninth prime is 23, which is quite large when we're multiplying).

```
In[103]:= With[{top = Times @@ Prime[Range[9]]}, Floor[n /. Solve[n^2 + n - 2 top == 0 && n > 0, n]]]

Out[103]= {21 122}
```

In[104]:=
```
Select[Select[Range[21 122], Not[PrimeQ[#]] && Not[PrimeQ[# + 1]] &],
  f[#] > 500 &, 1] // AbsoluteTiming
```
Out[104]=
```
{0.501923, {12 375}}
```

This turns out to correspond to the same answer:

In[105]:=
```
12 375 (12 376) / 2
```
Out[105]=
```
76 576 500
```